J. Symbolic Computation (1990) **9**, 503–513

Solution of a Hydrodynamic Lubrication Problem with Maple

R. M. CORLESS AND D. J. JEFFREY

Department of Applied Mathematics, University of Western Ontario, London, Ontario, Canada N6A 5B9

(*Received* 11 *May* 1987)

A set of partial differential equations, arising in a calculation of hydrodynamic lubrication effects, was solved using a perturbation technique. All of the algebraic manipulations required to find the solution were performed using Maple. The main challenge was the efficient handling and simplification of very long expressions, which was met by the power of Maple's built-in procedures and by algebraic transformations suggested by the solution to the lowest-order approximation. As a result, the solution was obtained to a higher order, with greater reliability, than would otherwise have been possible.

1. Introduction

The friction between moving bodies is greatly reduced by lubrication, when the surfaces that would be in contact are separated by a thin film of viscous fluid. The theory of lubrication is based on an analysis of the stress in this fluid film. The Navier–Stokes equations, which govern the fluid motion in the film, are greatly simplified by the fact that the film is very thin, because many physical effects, such as the inertia of the fluid, can be ignored. A typical approximate calculation of the force between two plane surfaces is given in Batchelor (1967, p. 219). Here we give a brief report on the use of the symbolic manipulation language Maple (Char *et al.*, 1988) to calculate the force between two spherical surfaces. For a detailed report, concentrating on the fluid mechanical theory, see Corless & Jeffrey (1988*a*). For a comparison of Maple, muMATH (Stoutemeyer & Rich, 1983), and CAMAL (Fitch, 1983) as aids in solving this problem, see Corless & Jeffrey (1988*b*). The notation of the present paper agrees with Corless & Jeffrey (1988*a*), though there are some differences with their second paper (1988*b*).

Let the gap between the moving surfaces be of width ε, a small quantity. The mathematical formulation uses the smallness of ε to convert the problem into a calculation of perturbation expansions for the fluid velocities and the pressure (O'Neill & Majumdar, 1970). The equations obtained after approximation are relatively simple, and the main difficulty in solving them is that, at each stage of the solution, the length of the expressions appears to grow exponentially. Since we are not interested in the full details of the flow field, but only some integral properties of it—the force, couple, and a quantity called the stresslet—the final expressions we require are not very long. As an added simplification, we do not want complete expressions for these quantities, but only the singular terms (those containing $\ln \varepsilon$ terms) in the expressions for the force, couple, and stresslet. Early investigators, working unaided by computer algebra, arrived at expressions for the singular terms by working always with asymptotic expressions. This is what was done in O'Neill & Stewartson (1967), for example, although only for a geometrically simpler case, and to a lower order. It is conceivable that something similar could have

0747–7171/90/040503+11 $03.00/0

been done for the present calculations, but it would have been a very long calculation, and the approach taken here is superior in that we have many additional checks that the differential equations and boundary conditions have been satisfied at each order of approximation. This method is also more efficient, because there are several cases to consider—for example, the spheres could be either translating or rotating. Unaided, a person would take roughly the same amount of time for each case, but with Maple new cases were done quickly.

The equations are expressed in cylindrical coordinates (r, θ, z) and must be solved for three velocity components u, v, w and a pressure p. It happens that θ can be removed from the equations because the velocity vector can be expressed as $(U\cos\theta, V\sin\theta, W\cos\theta)$ and so using also the approximation that the separation ε is a small quantity, we arrive at the algebraic problem summarized in O'Neill & Majumdar (1970) and Jeffrey & Onishi (1984).

The velocities and pressure are calculated using the series approximations

$$U = U_0 + \varepsilon U_1 + \varepsilon^2 U_2 + \ldots, \qquad V = V_0 + \varepsilon V_1 + \varepsilon^2 V_2 + \ldots,$$
$$W = \varepsilon^{1/2}(W_0 + \varepsilon W_1 + \varepsilon^2 W_2 + \ldots), \quad P = \varepsilon^{-3/2}(P_0 + \varepsilon P_1 + \varepsilon^2 P_2 + \ldots),$$

and the "stretched" variables $R = r/\varepsilon^{1/2}$ and $Z = z/\varepsilon$. The governing equations can be written as

$$\frac{\partial P}{\partial Z} = \varepsilon \frac{\partial^2 W}{\partial Z^2} + \varepsilon^2 \left(\Upsilon W - \frac{W}{R^2} \right), \tag{1.1}$$

$$\frac{\partial P}{\partial R} = \frac{\partial^2 U}{\partial Z^2} + \varepsilon \left(\Upsilon U - \frac{2}{R^2}(U+V) \right), \tag{1.2}$$

$$-\frac{P}{R} = \frac{\partial^2 V}{\partial Z^2} + \varepsilon \left(\Upsilon V - \frac{2}{R^2}(U+V) \right), \tag{1.3}$$

$$\frac{\partial U}{\partial R} + \frac{1}{R}(U+V) + \frac{\partial W}{\partial Z} = 0, \tag{1.4}$$

using the operator

$$\Upsilon = \frac{\partial^2}{\partial R^2} + \frac{1}{R}\frac{\partial}{\partial R}.$$

The fluid is enclosed between one moving and one stationary spherical boundary, and boundary conditions must be applied at each surface, where the velocities are known. The moving surface is a sphere of radius 1, centred at $Z = 1+\varepsilon$, and the stationary surface is a sphere of radius $-1/\kappa$, where $\kappa < 0$, centred at $Z = 1/\kappa$. The equations for the surfaces are expanded in the neighbourhood of the gap as series in ε. The moving surface is approximated by

$$Z_1 = (1 + \tfrac{1}{2}R^2) + \varepsilon \tfrac{1}{8}R^4 + \ldots = H_1 + \tfrac{1}{8}R^4 + \ldots$$

and the stationary surface by

$$Z_2 = \tfrac{1}{2}\kappa R^2 + \varepsilon \tfrac{1}{8}\kappa^3 R^4 + \ldots = H_2 + \varepsilon \tfrac{1}{8}\kappa^3 R^4 + \ldots.$$

If the moving sphere is rotating, we specify the velocities on the surface to be

$$U = -V = -1 - \varepsilon(Z-1) \quad \text{and} \quad W = -\mathrm{R} \qquad \text{on} \quad Z = Z_1.$$

If the moving sphere is translating, we have

$$U = -V = 1 \quad \text{and} \quad W = 0 \qquad \text{on} \quad Z = Z_1.$$

In either case, the velocities are zero on $Z = Z_2$, the stationary sphere.

2. Solution of the Equations with Maple

We used Maple version 4.0 (Char *et al.*, 1988), and later version 4.1, to solve this problem for each set of boundary conditions, with an error $O(\varepsilon^3)$. By the time this paper reaches publication, Maple version 4.3 will have been released. We remark that several procedures we wrote to solve this perturbation problem would have been coded differently had any of the later versions been available at the time this work was done. Some of the Maple code we did use is appended to the paper, as we used it at the time, together with some commentary on how it could be changed to use the features of the newer versions.

The system of equations is effectively decoupled at each order, and turns out to have solutions polynomial in Z and rational in R. While describing our general method, we shall include for illustration at each step specific expressions for the simplest (order 0) case. At each order, we first integrate equation (1.1) with respect to Z to obtain an expression for the pressure, which then contains an unknown function of R. At order zero, the unknown function is the only contribution, so $P_0 = P(R)$. Boundary conditions cannot be applied yet, but later they will lead to a differential equation for the unknown function introduced. Next, equations (1.2) and (1.3) can be integrated with respect to Z separately; for the zeroth-order solution, we get

$$U_0 = \frac{1}{2}\frac{\mathrm{d}P}{\mathrm{d}R} Z^2 + A(R)Z + B(R) \tag{2.1}$$

and

$$V_0 = \frac{1}{2}\frac{P}{R} Z^2 + C(R)Z + D(R), \tag{2.2}$$

where A, B, C, and D are unknown functions of R, which will be determined by the boundary conditions on U and V. For example,

$$\begin{bmatrix} A \\ B \end{bmatrix} = \begin{bmatrix} H_1 & 1 \\ H_2 & 1 \end{bmatrix}^{-1} \begin{bmatrix} -1 - \dfrac{1}{2}\dfrac{\mathrm{d}P}{\mathrm{d}R} H_1^2 \\ -\dfrac{1}{2}\dfrac{\mathrm{d}P}{\mathrm{d}R} H_2^2 \end{bmatrix} = h(R) \begin{bmatrix} -1 - \dfrac{1}{2}\dfrac{\mathrm{d}P}{\mathrm{d}R}(H_1^2 - H_2^2) \\ H_2 + \dfrac{1}{2}\dfrac{\mathrm{d}P}{\mathrm{d}R}(H_1 - H_2)H_1 H_2 \end{bmatrix}, \tag{2.3}$$

where

$$h(R) = \frac{1}{1 + QR^2} \quad \text{and} \quad Q = \tfrac{1}{2}(1 - \kappa)$$

are variables used to keep the length of the expressions manageable. The conservation equation (1.4) is then integrated with respect to Z to determine the W_i, and one of the boundary conditions on W is used to evaluate the unknown function of R that arises as a

constant of integration. This leaves us with an expression for W_i which depends on Z, R, and the still unknown $P(R)$. The resulting expression for W_0 is

$$\begin{aligned} W_0 = &\frac{1}{48}\frac{\mathrm{d}^2P(R)}{\mathrm{d}R^2}(-2Z+3+R^2+QR^2)(-R^2+2QR^2+2Z)^2 \\ &+\frac{1}{48}\frac{\mathrm{d}P(R)}{\mathrm{d}R}\frac{(-R^2+2QR^2+2Z)}{R}(14R^4Q^2+7QR^4-7R^4 \\ &-14QR^2Z+16R^2Z+30QR^2-15R^2-4Z^2+6Z) \\ &-\tfrac{1}{48}P(R)\frac{(-2Z+3+R^2+QR^2)(-R^2+2QR^2+2Z)^2}{R^2} \\ &+\tfrac{1}{4}R(-R^2+2QR^2+2Z)(-2QZ-2+4Q-QR^2+2R^2Q^2)h^2(R). \end{aligned} \tag{2.4}$$

The other boundary condition on W is used to determine an inhomogeneous linear second-order ordinary differential equation with rational coefficients for $P(R)$. It turns out that this differential equation differs from order to order only in the inhomogeneous terms; and further that only its particular integral is needed, because it can be shown by considering how the solution must decay at the edge of the gap that the general solution to the homogeneous equation cannot enter our solution—hence the constants of integration must be zero (O'Neill & Stewartson, 1967). At the zeroth order, this differential equation is

$$R^2\frac{\mathrm{d}^2P}{\mathrm{d}R^2}+h(R)R(1+7QR^2)\frac{\mathrm{d}P}{\mathrm{d}R}-P=-12QR^3h^3(R)$$

and it can easily be shown that a particular solution to this differential equation is given by

$$P(R)=\tfrac{6}{5}Rh^2(R).$$

In general, we find a solution to this ODE by looking for a rational solution; in fact, we can predict a solution of the form (polynomial in R) multiplied by $h^{n+2}(R)$ (which we recall is rational in R) at the nth order. To find the polynomial we generate a polynomial template of high enough degree, and use the Maple routines *map*, *coeff*, and *solve*. The resulting linear system for the unknown coefficients is overdetermined, and will have a unique solution only if our predicted form is correct. The Maple routine *solve* is capable of solving overdetermined linear systems (if they have solutions) and reporting "no solutions found" if the system is inconsistent, which would here require raising the degree of the template.

After the solution for U, V, W and P was obtained to the desired order, we wished to calculate the force, couple, and stresslet. We used Maple *tables* for the stress tensor and the vector normal to the spheres, and calculated the desired integrands, as follows. The desired coefficients of the stress tensor are calculated using

$$\varepsilon\sigma_{RZ}=\frac{\partial U}{\partial Z}+\varepsilon\frac{\partial W}{\partial R},$$

$$\varepsilon\sigma_{RR}=\varepsilon^{-1/2}P+2\varepsilon^{1/2}\frac{\partial U}{\partial R},$$

$$\varepsilon\sigma_{\theta Z}=\frac{\partial V}{\partial Z}-\varepsilon\frac{W}{R},$$

$$\varepsilon\sigma_{\theta R}=\varepsilon^{1/2}R\frac{\partial}{\partial R}(V/R)-\varepsilon^{1/2}\frac{U}{R}$$

and

$$\varepsilon\sigma_{ZZ} = -\varepsilon^{-1/2}P + 2\varepsilon^{1/2}\frac{\partial W}{\partial Z}.$$

The vectors normal to the surface of the spheres are, on the moving sphere,

$$n_Z = -1 + \varepsilon(Z-1), \qquad n_R = \varepsilon^{1/2}R,$$

and on the stationary sphere

$$n_Z = 1 - \varepsilon\kappa Z, \qquad n_R = -\kappa\varepsilon^{1/2}R,$$

and finally, the desired integrands for force, couple, and stresslet on the moving sphere are

$$\text{Force} = \mu\pi \int (\sigma_{RZ}n_Z + \sigma_{RR}n_R - \sigma_{\theta Z}n_Z - \sigma_{\theta R}n_R)\varepsilon \frac{\mathrm{d}n_R}{\mathrm{d}R}\,\mathrm{d}R,$$

$$\text{Couple} = \tfrac{1}{4}\pi \int \{n_R(\sigma_{ZR}n_R + \sigma_{ZZ}n_Z) - n_Z[(\sigma_{RR}n_R + \sigma_{RZ}n_Z) - (\sigma_{\theta R}n_R + \sigma_{\theta Z}n_Z)]\} \frac{\mathrm{d}n_Z}{\mathrm{d}R}\,\mathrm{d}R,$$

$$\text{Stresslet} = \tfrac{1}{4}\pi \int \{n_R(\sigma_{ZR}n_R + \sigma_{ZZ}n_Z) + n_Z[(\sigma_{RR}n_R + \sigma_{RZ}n_Z) - (\sigma_{\theta R}n_R + \sigma_{\theta Z}n_Z)]\} \frac{\mathrm{d}n_Z}{\mathrm{d}R}\,\mathrm{d}R,$$

and similarly for the stationary sphere.

These integrals are not actually calculated; the integrands are expanded in asymptotic series in powers of R as $R \to \infty$, and the $1/R$ term gives the desired singular behaviour. Notice that this term gives rise to a logarithmic singularity when integrated: $\ln R \to \infty$ as $R \to \infty$. The correct mathematical treatment of this term is given in Corless & Jeffrey (1988*a*). The manipulations needed in this calculation were merely multiplication, substitution of the value of Z on the appropriate boundary, and taking Taylor series with respect to ε and to $1/R$. We remark that, due to our training on the older computer algebra systems CAMAL (Fitch, 1983) and REDUCE (Hearn, 1984), we expected that a feature like the "let EPS**3 = 0" construct, which automatically eliminates higher powers of EPS during calculation, would be useful here. Maple does not explicitly have such a construct; however, we are assured by M. B. Monagan, one of the designers of Maple, that the Maple routine *taylor* automatically avoids calculation of higher-order products if it can. So, if one is careful to code products of truncated power series entirely inside calls to *taylor*, one should be able to save significantly on computing expense. However, explicit calls to *expand* first would force the calculation of these unnecessary terms, and should be avoided.

3. Verification of Results

Although it is often claimed for computer algebra systems that they perform algebra more accurately than a human, it must be remembered that in an application of the size described here the possibility of algebra errors has been replaced by the possibility of programming errors. So, the results were checked in four ways. First, the solutions obtained were substituted back into the original differential equations and boundary conditions, which were indeed satisfied to the proper order of ε. Secondly, selected terms of the solution at the zeroth and first orders were compared with a solution earlier produced by CAMAL (Jeffrey & Onishi, 1984). Thirdly, the quantities of interest—the force, couple, and stresslet—were compared for internal consistency using a reciprocal theorem from fluid mechanics: the force on the moving sphere in the rotational case should be the same as the couple on the moving sphere in the translation case. Finally, a comparison was

Table 1. Force, couple, and stresslet on the moving sphere (rotational case)

Force	$\frac{(4\kappa-1)}{5(\kappa-1)^2}\ln\varepsilon-\frac{32\kappa^3+33\kappa^2+83\kappa-43}{125(\kappa-1)^3}\varepsilon\ln\varepsilon$ $-\frac{6844\kappa^5-78947\kappa^4-309572\kappa^3-167647\kappa^2+80828\kappa+29069}{393750(\kappa-1)^4}\varepsilon^2\ln\varepsilon+O(\varepsilon^3\ln\varepsilon)$
Couple	$-\frac{2}{5(\kappa-1)}\ln\varepsilon+\frac{2(8\kappa^2-6\kappa+33)}{125(\kappa-1)^2}\varepsilon\ln\varepsilon$ $+\frac{(1711\kappa^4-1804\kappa^3-79614\kappa^2-64279\kappa-41864)}{196875(\kappa-1)^3}\varepsilon^2\ln\varepsilon+O(\varepsilon^3\ln\varepsilon)$
Stresslet	$-\frac{(2\kappa+1)}{10(\kappa-1)^2}\ln\varepsilon+\frac{(16\kappa^3+61\kappa^2+180\kappa-2)}{250(\kappa-1)^3}\varepsilon\ln\varepsilon$ $+\frac{(3422\kappa^5-71917\kappa^4-621727\kappa^3-765317\kappa^2-316802\kappa+80791)}{787500(\kappa-1)^4}\varepsilon^2\ln\varepsilon+O(\varepsilon^3\ln\varepsilon)$

made with a solution valid when the gap is not small, which is found as a power series (Jeffrey, 1989). The singularities calculated here can be used to improve the convergence of the power series solution, but obviously only if they are correct (Corless & Jeffrey, 1988*a*). Table 1 shows a sample of the results; as is evident, the final expressions are not very long, although the intermediate expressions for the velocities and pressure occupied several pages of computer output.

4. Conclusions

With the calculation of the force, couple and stresslet to a higher power of ε than had been previously attempted, Maple has shown itself to be a useful tool for lubrication theory, and more generally for perturbation expansions. The use of computer algebra for perturbation expansions is hardly new, but has generally been confined to the older languages. In particular, CAMAL was designed with this sort of problem in mind, and in contrast, Maple was designed to be a more general symbolic manipulation language. This report confirms that perturbation series have not been sacrificed by the designers of Maple in their quest for more generality. In fact, the generality of Maple added greatly to the convenience of using Maple for the solution of this problem. For example, CAMAL has no GCD algorithm, and even the division of polynomials could be problematic if the divisor had more than two terms; here, one need not worry about this.

Several Maple functions were particularly useful for this problem: *normal*, *taylor* and *solve*, together which accounted for most of the computer resources used. *Normal* was used to simplify the rational expressions that occurred, and had a very noticeable effect on the size of the expressions. *Taylor* was necessary to drop higher order terms in ε as well as to isolate the coefficient of $1/R$ in the integrands for the force, couple and stresslet. *Solve* was useful for the solution of the ordinary differential equation at each stage; once the polynomial form for the solution had been determined by inspection, a call to *solve* to solve the overdetermined set of linear equations was convenient and effective. In addition, the Maple programming language was very useful in setting up some auxiliary routines for simplification.

On the other side of the coin, there are a few features that we missed in Maple version 4.0. We are happy to report that the main conveniences have been put in place by Maple version 4.3—the most notable additions being the new manual for Maple version 4.2 and the new "echo" facility of version 4.3, that allows echoing of input commands together with the output. This facility is invaluable in a project of this size, and the advent of the Macintosh version of Maple, with its windowing capability, does not obviate the need for echoing. There are still some useful things missing, however; for example, a facility allowing Maple to quit reading a file if it encounters an error (currently, Maple accommodates its test suite by continuing to read from the file even after an error has been encountered, which is appropriate only in a test suite context). Also, it would be nice to have a feature that would allow a user to send a break to Maple if the expression being printed was too long. However, the final conclusion is that in spite of missing features, even Maple 4.0 was powerful enough to solve this sizeable problem.

References

Batchelor, G. K. (1967). *Introduction to Fluid Dynamics*. Cambridge: Cambridge University Press.

Char, B. W., Geddes, K. O., Gonnet, G. H., Monagan, M. B., Watt, S. M. (1988). *The Maple Reference Manual, (5th Edn)*. Waterloo, Canada: WATCOM Publications.

Corless, R. M., Jeffrey, D. J. (1988*a*). Stress moments of nearly touching spheres in low Reynolds number flow. *Z. angew. Math. Phys.* **39,** 874–884.

Corless, R. M., Jeffrey, D. J. (1988*b*). A comparison of three computer algebra languages with a problem in hydrodynamic lubrication theory. *SIGSAM Bulletin* **22,** no. 2, Issue no. 84.

Fitch, J. P. (1983). *CAMAL User's Manual*, 2nd edn. University of Cambridge Computer Laboratory, Cambridge, UK.

Gonnet, G. H. (1986). An implementation of operators for symbolic algebra systems. *Proc. SYMSAC* **86,** 239–243. Waterloo, Canada.

Hearn, A. C. (1984). *REDUCE User's Manual: Version 3.1*. Santa Monica, CA: The Rand Corporation.

Jeffrey, D. J. (1989). Interactions between two equal spheres at low Reynolds numbers. *Utilitas Mathematica* (in press).

Jeffrey, D. J., Onishi, Y. (1984). The forces and couples acting on two nearly touching spheres in low-Reynolds-number flow. *Z. angew. Math. Phys.* **35,** 634–641.

O'Neill, M. E., Stewartson, K. (1967). On the slow motion of a sphere parallel to a nearby plane wall. *J. Fluid Mech.* **27,** 705–724.

O'Neill, M. E., Majumdar, S. R. (1970). Asymmetrical viscous fluid motions caused by the translation of rotation of two spheres. Part II: Asymptotic forms of the solutions when the minimum clearance between the spheres approaches zero. *Z. angew. Math. Phys.* **21,** 180–187.

Stoutemeyer, D. R., Rich, A. (1983). *muMATH-83 Reference Manual*. Honolulu, HA: The Soft Warehouse.

Appendix—Selected Version 4.0 Maple Procedures

BASIC DEFINITIONS AND AUXILIARY ROUTINES

```
#The boundary conditions on sphere 1 are applied on the parabola Z=H1
H1:=1+(1/2)*R^2;
#The boundary conditions on sphere 2 are applied on the parabola Z=H2
H2:=(1/2)*KAPPA*R^2;
#Experience with CAMAL shows that Q is a better variable to work in.
KAPPA:=1-2*Q;
#The total distance between the parabolas is H (KAPPA is negative)
H:=expand(H1-H2);
#
#Also following the CAMAL program, we introduce h=1/H.
```

```
#We use the user interface of the function diff,
#to define the derivative of h by defining a function `diff/h`.
#
`diff/h`:=proc(exp,indet)
  -2*Q*exp*h(exp)^2*diff(exp,indet)
end;
#The upsilon operator is defined in the paper Corless & Jeffrey (1988a)
UPS:=proc(exp)
  diff(R*diff(exp,R),R)/R;
end;
#
#This matrix is used to solve the equations resulting from applying the b.c.
#
BCmatrix:=array([[1,-1],[-H2,H1]]):
BCmatrix:=linalg[scalarmul](BCmatrix,h(R)):
#
#This routine attempts to divide the numerator by H and reduce the highest
#power of h(R) by one. Mainly needed for muMATH but retained for maple
ReduceLast:=proc(exp)
  local lexp,lterm,quot,deg;
  lexp:=collect(convert(exp,polynom),h(R));
  deg:=degree(lexp,h(R));
  lterm:=lcoeff(lexp,h(R));
  if not divide(lterm,H,quot) then
    lexp
  else
    lexp-lterm*h(R)^deg+quot*h(R)^(deg-1)
  fi;
end:
#
#This routine uses the properties of h(R) to put several terms over a
#common denominator (i.e. they are all multiplied by the same power of h)
comden:=proc(exp)
  local lexp,deg,result,i;
  lexp:=collect(convert(exp,polynom),h(R));
  deg:=degree(lexp,h(R));
  result:=0;
  for i from 0 to deg do
    result:=result+coeff(lexp,h(R),i)*H^(deg-i);
  od;
  expand(result)*h(R)^deg;
end:
#
#The aim of this routine is to use taylor to remove higher order terms, but
#otherwise avoid the taylor data type which caused problems for expand
#in Maple version 4.0.
#Notice that the variable Order will determine how many terms are kept.
taylEPS:=proc(exp)
  convert(taylor(exp,EPS=0),polynom);
end:
#The two surfaces are given by Z=Z1 and Z=Z2
#
Z1:=1+(1-(1-EPS*R^2)^(1/2))/EPS:
#>Order:=4; for taylEPS above
Order:=4;
Z1:=taylEPS(Z1);
#
Z2:=(1-(1-EPS*KAPPA^2*R^2)^(1/2))/(EPS*KAPPA):
```

```
Z2:=taylEPS(Z2);
#Laplace is the scaled laplacian operator.
LAPLACE:=proc(exp):
  diff(diff(exp,Z),Z)+EPS*diff(R*diff(exp,R),R)/R:
end:
#A little routine here to extract terms
#Supplanted already in version 4.1 by the routine "isolate"
extract:=proc(exp,subexp):
  expand(exp-subs(subexp=0,exp)):
end:
```

MAPLE CODE FOR THE ZEROTH ORDER

```
readlib(isolate):
#Program in Maple to solve asymmetric motion between spheres in low Reynolds
#number flow (lubrication theory). #December 87 Rob Corless & David Jeffrey
#
#The following functional references can be streamlined with the use of
#the (new for version 4.3) facility "alias", which can serve as a partial
#replacement for the DEPENDS feature of muMATH, REDUCE, or Macsyma
#
MAXORDER:=2:
for i from 0 to MAXORDER do
  p.i:=P.i(R,Z):
  u.i:=U.i(R,Z):
  v.i:=V.i(R,Z):
  w.i:=W.i(R,Z):
od:
#
#The solution of the first (trivial) equation.
p0:=P0(R):
#expand each function in powers of EPSilon
#
i:='i':
p:=sum('p.i'*EPS^i,i=0..MAXORDER):
u:=sum('u.i'*EPS^i,i=0..MAXORDER):
v:=sum('v.i'*EPS^i,i=0..MAXORDER):
w:=sum('w.i'*EPS^i,i=0..MAXORDER):
EQuu:=-R^2*diff(p,R)+R^2*diff(u,Z,Z)+EPS*(R^2*UPS(u)-2*(u+v)):
EQuv:=p*R+R^2*diff(v,Z,Z)+EPS*(R^2*UPS(v)-2*(u+v)):
EQuw:=-diff(p,Z)+EPS*diff(w,Z,Z)+EPS^2*(UPS(w)-w/R^2):
EQuC:=R*diff(u,R)+u+v+R*diff(w,Z):
#Equations now established. Separate orders
#
EQuuS:=collect (EQuu,EPS):
EQuvS:=collect (EQuv,EPS):
EQuwS:=collect (EQuw,EPS):
EQuCS:=collect (EQuC,EPS):
#we have now separated all equations by order in EPS. Now we integrate.
#
int(int(coeff(EQuuS,EPS,0),Z),Z):
#Now we have reduced it to an algebraic equation.
u0:=op(2,isolate(",U0(R,Z)));
#This gives us Equation (2.1) in the text.
#Of course we will need to use the boundary conditions to evaluate the
#arbitrary function, linear in Z, which should be added to the above.
#
#Now we do the same for the equation for v0
int(int(coeff(EQuvS,EPS,0),Z),Z):
```

```
v0:=op(2,isolate(",V0(R,Z)));
#This gives us equation (2.2) in the text, once constants of integration
#are added.
#
#The boundary conditions are U0(R,Z)=-1 when Z=H1, U0=0 when Z=H2.
#
sol1:=subs(Z=H1,u0+1):
sol2:=subs(Z=H2,uO):
soln:=linalg[multiply](BCmatrix,[-sol1,-sol2]):
#soln[1]⇔A(R) in the text
#soln[S]⇔B(R) in the text, equation (2.2)
u0:=u0+soln[1]*Z+soln[2]:
u0:=comden(u0);
#This gives us the results presented in equation (2.3) in the text
#V0=1 when Z=H1, and V0=0 when Z=H2.
#
sol1:=subs(Z=H1,v0-1):
sol2:=subs(Z=H2,v0):
soln:=linalg[multiply](BCmatrix,[-sol1,-sol2]):
#soln[1]⇔C(R)
#soln[2]⇔D(R) in the text, equation (2.3)
v0:=v0+soln[1]*Z+soln[2]:
v0:=comden(v0);
#
#Now we find w0.
#
(u0+v0+R*diff(u0,R)):
w0:=-(1/R)*int(",Z):
#
#This next construction evaluates the arbitrary function of R
#by applying the boundary condition w0=0 at Z=H2.
#
w0:=w0-subs(Z=H2,w0):
#At this point we have the result presented in equation (2.4)
#in the text, though several simplifying collections and factorings
#were done to get it in the form presented in the text.
#
#we now obtain the differential equation for p0
#
subs(Z=H1,w0)+R:
difeq:=12*R^2*":
difeq:=comden(difeq):
difeq:=normal(difeq/(H*h(R)));
#Finally we substitute the known solution for p0
#
sol:=(6/5)*R*h(R)^2;
expand(subs(P0(R)=sol,difeq));
difeq:=comden(");
#
#Having verified that p0 is (6/5) R h(R)^2
#we substitute this into the expressions for u0, v0 and
#
#w0 to obtain explicit expressions for the zeroth solution.
#
#Note that the following substitutions can be carried out with a
#single assignment statement in version 4.1 or later.
#
p0:=sol:
```

```
u0:=subs(P0(R)=sol,u0):
v0:=subs(P0(R)=sol,v0):
w0:=subs(P0(R)=sol,w0):
u0:=ReduceLast(u0,h(R),H):
u0:=normal(comden(u0));
v0:=ReduceLast(v0):
v0:=normal(comden(v0));
w0:=ReduceLast(w0):
w0:=normal(comden(w0));
```

SOLUTION OF THE ODE FOR P(R) AT SECOND ORDER

```
#
#Now we proceed to solve this differential equation. We
#expect a solution of the form S(R)/H^6, where S(R) is a
#polynomial.
subs(P2(R)=S(R)*h(R)^6,normal(difeq/h(R)^5)):
numer(''):
polydif:=comden('',h(R),H)/h(R)^8;
i:='i':
#
#The following construction was necessary for version 4.0 because of a
#subtle bug in the Maple routine "sum". The sum function behaved differently
#for large integer arguments than it did for small ones. There was an
#obscure bug with this process in earlier versions of Maple. However,
#this workaround also works in the later versions, though it is no
#longer necessary.
#In order to avoid premature evaluation of a.i in the loop,
#we use quotes—but this shifts the index of i by one by
#reason of the delayed evaluation.
#
solution:=0;
for i from -1 to 14 do
  solution:=solution+'a.i'*R^(2*i+1)
od:
subs(S(R)=solution,polydif):
eqs:=collect(numer(''),R):
deg:=degree(eqs,R);
#Use "solve" to solve the overdetermined system of equations,
#and use "map" and "coeff" to conveniently set up the equations.
#
solve(map(proc(n)coeff(eqs,R,n)end,{$(0..deg)}),{a.(0..15)});
assign(''):
#
solution:=collect(expand(solution),R);
p2:=subs(P2(R)=solution*h(R)^6,p2):
```